
LES DONNÉES CHRONOLOGIQUES SONT PARTOUT, COMMENT UTILISER 

LEURS POTENTIELS AVEC SAS 9.4 ? 

 

Aujourd’hui la plupart des données dont nous disposons 
possèdent une composante temporelle. Nous allons donc 
développer dans cet article les principales procédures du 
module SAS/ETS®, version 12.3 (Econometrics and Time 
Series), disponible avec SAS 9.4 et qui regroupe toutes 
les procédures permettant de tirer parti de cette 
information sur le temps. Nous nous attarderons sur la 

spécificité des procédures les plus usuelles et les plus utiles, leurs capacités ainsi que les nouveautés 
offertes par cette dernière version. 

1. LA PROCÉDURE EXPAND OU COMMENT CONSOLIDER SES DONNÉES 

1.1. Théorie 

Avant de commencer à estimer des modèles et à utiliser réellement la puissance des séries 
temporelles, il faut regarder la forme qu’elles ont et leur contenu. Les deux premières procédures 
présentées dans cet article s’attachent à vous aider à comprendre vos données et à les rendre de 
meilleure qualité pour accroître la pertinence des résultats en sortie de vos analyses. 
 

Ainsi, la procédure EXPAND est la première à découvrir (ou à redécouvrir) dans le module SAS/ETS. 
En effet, elle vous permet de changer la fréquence de votre table. Dans l’instruction PROC EXPAND 
vous devez spécifier la fréquence de base (FROM=) et la fréquence cible (TO=). Vous pouvez donc 
passer des données de la maille « mois » à la maille « trimestre » ou « année ». Dans ce sens-là, de 
la maille fine vers la maille globale, il suffit d’agréger, cela peut être fait simplement sans la 
procédure. En revanche, dans le sens inverse, de la maille globale vers la maille fine, cela devient 

plus complexe et l’utilisation de la procédure EXPAND prend tout son sens. 

 
Pour indiquer à SAS la technique de changement de fréquence, vous devez utiliser l’instruction 
CONVERT qui dispose des options suivantes : 

 METHOD= (peut prendre les valeurs: SPLINE, JOIN, STEP, AGGREGATE et NONE) 
permettant de définir la méthode de conversion d’une fréquence vers l’autre. 

 OBSERVED= (peut prendre les valeurs: TOTAL, AVERAGE, BEGINNING, MIDDLE, et END) 

définit les caractéristiques des observations des séries en entrée et en sortie. La nature des 
séries modifiera le traitement des données par la suite. 

 
L’autre possibilité offerte par cette procédure est l’imputation des valeurs manquantes. Son utilisation 
est simple, il suffit de retirer l’option TO= dans l’instruction PROC EXPAND et de choisir les options 
voulues dans l’instruction CONVERT et la procédure remplacera d’elle-même les valeurs manquantes 
de votre série de la manière indiquée. 

 

1.2. Exemple d’application 

L’exemple est issu de la documentation officielle sur cette procédure. 
 
Dans une première table (qtrly), nous disposons de données trimestrielles et dans une seconde 

(monthly) de données mensuelles avec les valeurs représentées ci-dessous.  
 

                                                        
 

On utilise la procédure EXPAND pour passer la table qtrly à la maille mensuelle en précisant que les 

données sont des moyennes et en appliquant la méthode par défaut (SPLINE), puis nous fusionnons 
les deux tables : 
 

proc expand data=qtrly out=temp from=qtr to=month; 

   convert gdp gd / observed=average; 

   id date; 

run; 

http://www.sas.com/fr_fr/support.html

data combined; 

   merge monthly temp; 

   by date; 

run; 

 
Nous obtenons donc une table globale à la maille mensuelle : 
 

 

 
 
 
 
 
 

 

2. LES PROCÉDURES X11/X12 OU COMMENT SÉPARER LES TENDANCES 

2.1. Théorie 

Les procédures X11 et X12 sont basées sur les techniques d’ajustement saisonnier X11 et X12 de 
l’U.S. Bureau of Census. Etant donné qu’elles sont très similaires, nous n’évoquerons dans cet article 
que la plus utilisée, la procédure X12. 

 
Cette procédure ajuste des séries temporelles mensuelles et trimestrielles pour en extraire les 
composantes saisonnières, les tendances cycliques et les erreurs de fluctuations. La procédure 
dispose de nombreuses instructions et options mais nous allons nous concentrer sur les différentes 
sorties que nous pouvons obtenir. En effet, le principal avantage de celle-ci est qu’elle permet de 
créer des séries sans saisonnalité ou alors de séparer les différentes composantes d’une même série 
pour mieux appréhender cette dernière. 

 

2.2. Nouveautés 

Vous trouverez à la page « X12 Procedure » la liste des nouvelles instructions/options disponibles en 
version 12.1 et 12.3 du module SAS/ETS. Elles sont principalement axées sur des points très avancés 
que nous n’abordons pas dans cet article. 

 

2.3. Exemple d’application 

Voici un exemple simple de code qui vous permettra, en jouant sur la liste des tables dans l’instruction 
ODS SELECT, de voir les capacités de la procédure et les possibilités de sorties : 
 
data sales; 

   set sashelp.air; 

   sales = air; 

   date = intnx( 'month', '01sep78'd, _n_-1 ); 

   format date monyy.; 

run; 

 proc x12 data=sales date=date; 

   var sales; 

   x11; 

   ods select d1 d10d d11 d12 d13; 

run; 

 
Nous avons choisi ici un nombre restreint de tables mais il s’agit des plus généralement souhaitées. 
 

3. LA PROCÉDURE FORECAST OU LA PROCÉDURE ORIGINELLE 

3.1. Théorie 

Maintenant que nous avons une idée de nos données et qu’elles sont nettoyées, nous pouvons 
commencer à essayer de les « faire parler ». La procédure FORECAST est l’une des premières 
développées dans ce module, elle est donc ancienne mais efficace pour faire rapidement des analyses 

http://www.sas.com/fr_fr/support.html
http://support.sas.com/documentation/cdl/en/etsug/66100/HTML/default/viewer.htm#cx12fdisplayedoutput.htm

simples de manière automatique. Nous verrons par la suite des procédures « filles » qui sont conçues 

pour traiter des points spécifiques de la procédure FORECAST (comme les procédures ESM et 
AUTOREG). 
 
La syntaxe de la procédure est assez simple : 
 PROC FORECAST avec différentes options 

 les options de tables  
 DATA= définit la table à analyser 

 OUT= définit le nom de la table qui contiendra les prévisions 
 OUTEST= définit la table contenant les paramètres d’estimations et les statistiques  

 Les paramètres de prévision 
 INTERVAL= définit l’intervalle entre deux données (mois, …) 
 SEASONS= définit le rythme de la saisonnalité 
 LEAD= définit l’horizon de prévision 

 METHOD= définit la méthode d’estimation parmi les méthodes suivantes : 
 STEPAR : méthode autorégressive pas à pas 

 EXPO : méthode de lissage exponentiel 
 WINTERS : méthode de lissage Holt-Winters 
 ADDWINTERS : méthode de lissage Holt-Winters additif 

 ID permet d’identifier la variable de référence temporelle 
 VAR indique la liste des variables à prévoir 

 

3.2. Exemple d’application 

L’exemple repris est issu de la documentation officielle dont le code est : 
 
proc forecast data=sashelp.usecon interval=month 

              method=winters seasons=month lead=12 

              out=out outfull outresid outest=est; 

   id date; 

   var vehicles; 

   where date >= '1jan80'd; 

run; 

 
Les tables de sorties sont les suivantes respectivement OUT et EST (un extrait seulement de chacune 

est illustré ci-dessous): 
 

 
 
Dans la première table (OUT), les dates pour lesquelles les modalités ACTUAL/FORECAST/RESIDUAL 
sont présentes sont des données passées (avec les valeurs réelles, prévues et les résidus) alors que 
les dates avec FORECAST/L95/U95 sont des données prévues (valeurs prévues, borne inférieure et 
supérieure de l’intervalle de confiance à 95%) pour un horizon donné (valeur de _LEAD_, troisième 

colonne dans la sortie). La seconde table (EST) fournit des informations sur les paramètres du modèle 
et certaines statistiques. 
 

 
 

http://www.sas.com/fr_fr/support.html

4. LA PROCÉDURE ESM OU COMMENT FAIRE UN LISSAGE EXPONENTIEL 

4.1. Théorie 

La procédure ESM est la procédure incontournable pour faire du lissage exponentiel. Elle peut aussi 
bien en faire sur des séries temporelles que sur des données transactionnelles. Elle permet de faire 
un lissage : 
 Simple 
 Double 

 Linéaire 
 Damped-trend 
 Saisonnier 
 Holt-Winters (additif ou multiplicatif) 
 
Elle offre également des possibilités de transformation de la série : 

 Log 

 Racine Carrée 
 Logistique 
 Box-Cox 
 
La syntaxe est proche de celle de la procédure FORECAST puisque les options DATA=, LEAD=, OUT= 
et OUTEST= sont également présentes avec les mêmes fonctions/capacités. 

Cependant, plusieurs particularités : 
 Dans l’instruction PROC ESM, l’option PLOT= permet l’affichage de différents graphiques. 
 L’option INTERVAL= intervient dans l’instruction ID. 
 La méthode est spécifiée dans l’instruction FORECAST (équivalente à VAR dans la procédure 

FORECAST) au moyen des options MODEL= et TRANSFORM=. 
 

4.2. Exemple d’application 

Nous reprenons dans cette section le même exemple que pour la procédure FORECAST avec le code 
adapté : 
 
proc esm data=sashelp.usecon seasonality=12 lead=12 

   out=out outstat=stat outfor=for outest=est; 

 id date interval=month; 

 forecast vehicles/ model=winters; 

 where date >= '1jan80'd; 

run; 

 
Les tables de sorties sont toujours OUT et EST auxquelles s’ajoutent STAT et FOR: 
 

  

 
 

http://www.sas.com/fr_fr/support.html

Vous remarquerez que la première table, en haut à gauche, (OUT) est plus simple qu’avec la 

procédure FORECAST. Elle contient deux colonnes, la date et la valeur de la variable cible où les 
données historiques et de prévisions se confondent. La deuxième, en haut à droite, (EST) est 
également quelque peu différente sur la forme mais elle contient uniquement les paramètres 
d’estimation et des statistiques liées à ces derniers. La table centrale (STAT) contient toutes les 
statistiques d’ajustement par variable estimée (ici elle ne contient donc qu’une seule ligne). Les deux 
dernières images sont le début et la fin de la table FOR. Il s’agit en fait de la table de sortie la plus 
complète qui vous permettra d’avoir une bonne idée de la modélisation effectuée. 

 

5. LA PROCÉDURE AUTOREG OU COMMENT FAIRE DES ESTIMATIONS LORSQUE 

L’ON A DES ERREURS AUTOCORRÉLÉES OU HÉTÉROSCÉDASTIQUES 

5.1. Théorie 

Cette procédure AUTOREG permet l’estimation et la prévision de modèle linaire avec des erreurs 

autocorrélées ou hétéroscédastiques. Le modèle à erreurs autorégressives permet de corriger 
l’autocorrélation, et le modèle GARCH (Generalized AutoRegressive Conditional Heteroscedasticity) 
et ses dérivés de corriger l’hétéroscédasticité. Il s’agit de la seconde procédure « fille » de la 
procédure FORECAST pour la partie traitement de l’autocorrélation.  
Nous allons nous concentrer sur la partie la plus complexe adressée par cette procédure, la 

modélisation GARCH. En effet, la partie autocorrélation se règle facilement en ajoutant l’option 
NLAG=, avec les retards  adéquats, dans l’instruction MODEL où est définit notre modèle, pour que 
cette partie soit traitée. 
 
En revanche, les modèles de type GARCH sont plus complexes à comprendre et à faire comprendre 
à SAS. Nous allons détailler la syntaxe en nous attardant sur les particularités et les points importants 

de celle-ci : 
 PROC AUTOREG : l’option OUTEST= nom de la table de sortie est l’option la plus importante de 

cette instruction (avec l’option DATA= qui est obligatoire). En effet, nous verrons par la suite 
que sortir les paramètres de nos modèles de type GARCH est important. 

 MODEL : en plus des options classiques de définition d’un modèle, il faut porter attention à : 
 GARCH= permet de définir le modèle GARCH au moyen entre autres des options ci-dessous 

http://www.sas.com/fr_fr/support.html

 DIST= permet de définir la loi suivie par le terme d’erreur du modèle (T pour Student ou 

par défaut NORMAL) 
 TYPE= est l’option la plus importante permettant de définir le type de modèle GARCH à 

utiliser (EGARCH, TGARCH, IGARCH,…) 
 P= spécifie les ordres de la partie ARCH qui doivent être estimés 
 Q= spécifie les ordres de la partie GARCH qui doivent être estimés 

 
Ce type de modèle, GARCH, TGARCH et les autres modèles à erreurs hétéroscédastiques, peuvent, 

selon les techniques d’estimation et les lois suivies par le terme d’erreur, avoir du mal à converger, 
comme avec la procédure MODEL par exemple. De ce fait, les paramètres obtenus avec la procédure 
AUTOREG servent souvent de paramètres d’initialisation pour l’estimation d’un modèle légèrement 
différent avec la procédure MODEL que l’on décrira dans la section suivante. En effet,  comme nous 
le verrons cette autre procédure offre davantage de possibilités mais peine parfois à converger alors 
que la procédure AUTOREG converge beaucoup plus facilement mais les modèles sont plus simples. 

Ainsi, l’option OUTEST= est très importante et doit presque être un réflexe. En effet, cette procédure 
peut n’être qu’un intermédiaire pour la procédure MODEL. 

 

5.2. Nouveautés 

La version 12.1 du module SAS/ETS apporte quelques nouveautés dont: 

 Le passage en production des tests de racine unitaire ERS et Ng-Perron 
 Le support des tests de cointégration de Shin avec p-values 
 Le support de la matrice d’estimateur HAC (Heteroscedasticity- and Autocorrelation-Consistent) 
 Le support des tests de changements structuraux multiples de Bai et Perron  

 

5.3. Exemple d’application 

Voici un exemple de code possible avec cette procédure fourni par la documentation officielle :  
 
proc autoreg data=ibm_long outest=garch_family; 

   ar_1 :           model r = / noint nlag=1 method=ml; 

   arch_2 :         model r = / noint garch=(q=2); 

   garch_1_1 :      model r = / noint garch=(p=1,q=1); 

   st_garch_1_1 :   model r = / noint garch=(p=1,q=1,type=stationary); 

   ar_1_garch_1_1 : model r = / noint nlag=1 garch=(p=1,q=1); 

   igarch_1_1 :     model r = / noint garch=(p=1,q=1,type=integ,noint); 

   egarch_1_1 :     model r = / noint garch=(p=1,q=1,type=egarch); 

   qgarch_1_1 :     model r = / noint garch=(p=1,q=1,type=qgarch); 

   tgarch_1_1 :     model r = / noint garch=(p=1,q=1,type=tgarch); 

   pgarch_1_1 :     model r = / noint garch=(p=1,q=1,type=pgarch); 

run; 

 
Vous voyez dans le code les différents types de modèles GARCH et avec erreurs autorégressives qu’il 
est possible d’estimer. On y voit l’application de différents modèles, avec des composantes AR pour 

les erreurs, entre autres. 
A noter qu’au début de chaque ligne un nom est donné à chaque instruction MODEL au moyen de 

commandes du type « ar_1 : ». L’utilisation de ces commandes ne se limite pas à cette procédure.  
Ci-dessous, on retrouve une impression de la table des estimateurs des différents modèles (nommé 
avec les noms définis dans la procédure). 
 

http://www.sas.com/fr_fr/support.html
http://support.sas.com/documentation/cdl/en/etsug/66100/HTML/default/viewer.htm#etsug_autoreg_syntax07.htm

Nous ne détaillerons pas plus les résultats de cette procédure, les paramètres estimés étant le 

principal objectif. 
 

6. LA PROCÉDURE MODEL OU COMMENT CONSTRUIRE ET ESTIMER SON MODÈLE 

6.1. Théorie 

Cette procédure est très différente de celles présentées dans les autres sections.  

En effet, elle résout les modèles de relations entre les variables d’un système d’équations non 

linéaires. Son utilisation est multiple, et avec elle, nous pouvons donc : 
 Estimer des paramètres via différentes méthodes dont : 

 OLS (Ordinary Least Squares) : moindres carrées ordinaires 
 2SLS (Two Stage Least Squares) et 3SLS : moindres carrées en 2 étapes  
 GMM (Generalized method of moments) : méthode des moments généralisés 

 Simuler du Monte Carlo 

 Estimer et résoudre des systèmes d’équations différentielles 
 Spécifier et estimer la structure de la covariance des erreurs 
 
Puisque cette procédure est capable d’estimer les paramètres d’un système d’équations de manière 
simultanée, dans notre cadre d’étude des données temporelles, nous pouvons nous en servir, par 
exemple, pour estimer les modèles de types GARCH. Nous allons nous concentrer dans notre 
explication sur cet aspect de la procédure mais, au vu de ces capacités, vous comprenez qu’elle est 

en mesure d’estimer quasi tous les modèles dès que vous êtes à même de lui spécifier un système 
d’équations (même non linéaires) et une méthode d’estimation. 

 
Nous allons détailler une partie de la syntaxe seulement, car celle-ci contient plus de 30 instructions 
différentes, à savoir celle nous permettant d’estimer un modèle GARCH.  
Nous expliquons 3 instructions: 

 PARAMETERS : permet de déclarer la liste des paramètres du modèle et de façon optionnelle leur 
valeur initiale. Par défaut, sans valeur initiale, elles seront toutes initialisées à 0.0001. Les 
paramètres peuvent aussi être déclarés dans l’instruction FIT mais il est conseillé de le faire dans 
celle-ci. 

 ERRORMODEL : permet de spécifier la loi suivie par la distribution des résidus. La structure doit 
être : 
 le nom de la variable dépendante/endogène ou le nom de la forme générale du modèle puis 

un tilde (~) et enfin la distribution choisie avec ses paramètres : 
Y ~ NORMAL(1) indiquerait que les erreurs de la variable Y suivraient une loi normale de 
variance 1 

 FIT : indique les systèmes d’équations à estimer et les méthodes et options à utiliser pour ce 

faire : 
 METHOD= permet de spécifier la méthode de minimisation des itérations à utiliser (si 

l’algorithme ne converge pas, essayer avec l’autre méthode). Par défaut, GAUSS. 

http://www.sas.com/fr_fr/support.html

 Méthode d’estimation utilisée : la liste est longue et différente en fonction de l’utilisation de 

la procédure. Dans notre cas, FIML (Full Information Maximum Likelihood) est conseillée. 
 ITPRINT : affiche à chaque itération les paramètres estimés, la valeur de la fonction objective 

et le critère de convergence. 
 MAXITER= permet de modifier le nombre maximum d’itérations autorisées (par défaut 100). 

Il convient de l’augmenter pour aider l’algorithme à converger mais cela augmentera aussi 
les ressources utilisées. 

 ESTDATA= permet de spécifier une table dont les premières observations serviront de 

valeurs initiales pour tout ou partie des paramètres. Il peut s’agir par exemple de la table 
OUTEST issue de la procédure AUTOREG.  
 

Nous avons détaillé les principales instructions pour paramétrer l’estimation mais il faut encore écrire 
le système d’équations. Une équation s’écrit comme une instruction, donc pour faire simple, on écrit 
la formule (avec toutes les possibilités offertes dans les étapes DATA classiquement) et on termine 

par un point-virgule. Etant donné que nous nous concentrons sur les modèles de types GARCH, les 
erreurs sont hétéroscédastiques et peuvent prendre plusieurs formes. Pour écrire l’équation 

explicitant la structure de la covariance des erreurs, il faut utiliser la notation de type H.var avec var 
la variable d’équation. 
Ainsi si y s’écrit y = a*x+b, alors pour qualifier la structure de l’erreur de cette équation, on écrira 
h.y=…  
L’exemple dans la sous-section vous permettra de mieux voir comment la syntaxe s’articule. 

 
Les systèmes d’équations pouvant être très complexes, la convergence des algorithmes n’est pas 
garantie, il faudra donc jouer sur plusieurs points : le nombre d’itérations, les méthodes utilisées et 
les valeurs initiales des paramètres. Il est conseillé d’initialiser « intelligemment » les paramètres, 
et utiliser le résultat de la procédure AUTOREG est un bon moyen - mais pas efficace à tous les 
coups. La convergence dépend bien évidemment de vos données et de la qualité de celles-ci. 
 

6.2. Nouveautés 

Les principales nouveautés qui arrivent avec la version 12.1 du module SAS/ETS concernant cette 
procédure sont : 
 Trois nouvelles options de copules ont été rajoutées. Les simulations de Monte Carlo peuvent 

dorénavant utiliser les copules Archimédiennes CLAYTON, GUMBEL et FRANK 

 L'option OPTIMIZE a été ajoutée à l’instruction SOLVE pour permettre la simulation de modèles 
qui incluent des contraintes sur les variables de résolution dans le système d'équations du modèle 
de programme. Les limites supérieures et inférieures sur les variables de résolution peuvent être 
imposées en utilisant l'instruction BOUNDS et les contraintes linéaires ou non linéaires sur les 
fonctions des variables de résolution peuvent être imposées en utilisant l'instruction LIMITER. 

 

6.3. Exemple d’application 

Vous trouverez plusieurs exemples dans la documentation officielle pour illustrer différentes 
utilisations de cette procédure mais nous allons dans cette partie plutôt illustrer la syntaxe au moyen 
d’un exemple de code pour estimer un modèle GARCH(1,1) sous la loi GED (Generalized Error 

Distribution) : 

 
proc model data=sample; 

parameters _AH_0 _AH_1 _GH_1 v 2; 

rport=intercept; 

h.rport = _AH_0 + _AH_1*zlag(resid.rport**2) + _GH_1*zlag(h.rport); 

zt=resid.rport/sqrt(h.rport); 

lamba=sqrt((2**(-2/v)*gamma(1/v))/gamma(3/v)); 

l=log(v/lamba)-0.5*abs(zt/lamba)**v-(1+v**(-1))*log(2)-log(gamma(1/v))-

0.5*log(h.rport); 

l=-l; 

errormodel rport ~ General(l,v); 

fit rport /fiml estdata=garch_1_1normal itprint; 

quit ; 

 

La procédure MODEL travaillera sur la table SAMPLE (première ligne du code) et estimera 4 

paramètres _AH_0, _AH_1, _GH_1 et v (ligne 2 du programme), le dernier paramètre étant 

http://www.sas.com/fr_fr/support.html

initialisé à 2. Les lignes 3 à 9 servent à définir le système d’équation du modèle que la procédure 

doit résoudre (RESID. est une notation, comme H., pour exprimer la différence entre la valeur prévue 
de la variable et sa valeur actuelle réelle, pour plus de détails cf. documentation officielle). La ligne 
10 indique que les erreurs de l’équation RPORT suivent une loi GED de paramètres 1 et v (paramètre 
estimé). L’avant dernière ligne demande l’estimation de l’équation RPORT (et de ces dépendantes) 
selon la méthode FIML, en imprimant à chaque itération certaines informations et en se servant de 
la table garch_1_1normal pour initialiser les paramètres. Pour information, la table OUTEST est issue 
de l’estimation par la procédure AUTOREG d’un modèle GARCH(1,1) sous loi normale et contenant 

des valeurs estimées pour _AH_0, _AH_1_ et _GH_1. 
 

7. LA PROCÉDURE ARIMA OU COMMENT ESTIMER AVEC BOX-JENKINS 

7.1. Théorie 

Cette procédure, même si elle est évoquée en dernier, n’en est pas moins importante. En effet, la 

principale modélisation actuelle des séries temporelles se fait en utilisant la technique ARIMA 
développée par Box-Jenkins. La procédure ARIMA se base sur cette théorie modélisant la série par 
ses valeurs précédentes et la série des erreurs de modélisation associée. Nous allons donc évoquer 
la syntaxe de celle-ci et en même temps détailler les possibilités qu’elle offre.  
Mais avant, il faut retenir que cette procédure s’utilise de manière itérative : on identifie le modèle 
le plus performant, et seulement ensuite on estime et prévoit la série. Elle ne dispose pas d’outils de 
détection du meilleur modèle (comme peuvent le faire les procédures du module High-Performance 

Forecasting). A l’utilisation, vous devrez donc disposer de deux procédures ARIMA successives (du 
moins dans la partie de création du modèle).  
De plus, les données traitées dans cette partie doivent être à intervalle identique (« equally 
spaced »), il ne peut donc pas s’agir de données transactionnelles. Il est possible de passer par 
d’autres procédures pour préparer les données, comme, par exemple, la procédure EXPAND. 
La syntaxe est la suivante : 
 

 PROC ARIMA : il s’agit bien entendu d’une instruction obligatoire disposant de 3 options 

 DATA= indique le nom de la table qui sera utilisée 
 PLOTS permet de choisir les graphiques affichés après l’exécution de la procédure 
 Out= nom de la table de sortie 

 BY : instruction secondaire permettant de réaliser l’analyse en fonction de valeurs de différentes 
variables du BY 

 IDENTIFY : seconde instruction obligatoire, elle reprend le nom des variables à modéliser, les 
différenciations à effectuer ainsi que les statistiques d’aide à la décision à afficher. Il est conseillé 
de réaliser cette étape pour chaque série que vous voulez prévoir. Nous n’allons reprendre que 
les options principales de cette instruction. 
 VAR= nomme la variable qui contient la série à analyser. Il est possible de spécifier entre 

parenthèses les termes de différenciation de la série. Par exemple, Var=X(1) équivaut à 
travailler avec la série résultante de 𝑋𝑡 − 𝑋𝑡−1. 

 ESACF, MINIC et SCAN sont des options retournant des statistiques permettant de 

déterminer les meilleurs p et q des composantes AR(p) et MA(q). 
 P=(Pmin :Pmax) borne la recherche des critères précédents 

 Q= (Qmin :Qmax) borne la recherche des critères précédents 
 STATIONARITY= permet de tester la stationnarité de la série en fonction de plusieurs tests 

(ADF, Phillips-Perron,…) 
 ESTIMATE : indique, en particulier, les paramètres des composantes AR et MA à estimer ainsi 

que la méthode à utiliser 
 P spécifie la partie autorégressive du modèle 
 Q spécifie la partie « moyenne mobile » du modèle 
 METHOD= spécifie la méthode d’estimation 

 FORECAST : indique les options de prévisions du modèle précédemment estimé 
 LEAD= définit l’horizon de prévision 
 OUT= nomme la table de sortie contenant les prévisions 

La procédure offre d’autres options qui ne sont pas détaillées ici mais que vous pouvez retrouver 
dans la documentation officielle. 
 

7.2. Exemple d’application 

Voici un exemple s’appuyant sur une table d’exemple, SASHELP.AIR : 

proc arima data=sashelp.air; 

   identify var=air(1,12) esacf minic scan; 

   estimate p=1 q=(1)(12) noint method=ml; 

   forecast id=date out=out lead=24; 

quit; 

 

Vous trouverez ci-dessous un extrait des résultats obtenus en exécutant ce code : 

La première sortie donne des informations générales, entre autres, sur la variable analysée, les 

périodes de différenciation, le nombre d’observations utilisées et le nombre d’observations éliminées 
à cause de la différenciation. Cette sortie teste également si les résidus de la série différenciée sont 
des bruits blancs.  
La deuxième sortie est reprise dans le but d’illustrer les tables d’aide à la décision quant à la meilleure 
forme du modèle. Dans notre exemple, un modèle ARIMA (1,0) semble le meilleur (donc ARIMA 
(1,1,0) x SARIMA(0,1,0)12 puisque deux différenciations sont reprises.  
Enfin, la dernière sortie illustre les résultats de l’estimation des paramètres du modèle et donne la 

forme de celui-ci en distinguant différenciation, facteurs autorégressifs et facteurs « moyenne-
mobile ». Nous pouvons également voir le début des prévisions effectuées avec notre modèle et les 
intervalles de confiance associés. 
 

7.3. Pour aller plus loin 

Nous avons travaillé dans cette section sur des modèles univariés mais dans la réalité, nous 

travaillons généralement avec plusieurs séries qui sont liées entre elles et nous ne pouvons donc pas 
les prévoir de manière indépendante. Pour ce faire, il faut donc passer à de la modélisation 
multivariée. La procédure VARMAX, que nous ne détaillerons pas ici est faite pour cela. 
 

8. EN CAS DE PROBLÈME 

8.1. Les problèmes les plus fréquents 

Si au lancement de l’une de ces procédures vous rencontrez une erreur du type « PROCEDURE XXXX 
NOT FOUND », soit le module n’est pas sous licence, soit il n’est pas installé. Il convient donc de 
vérifier que : 

 le module est présent dans votre licence (en lançant le code « proc setinit ; run ; », le 
module SAS/ETS doit apparaître dans le journal). 

 [sur Windows], un répertoire « ets » est présent dans le dossier d’installation (par défaut 
« C:\Program Files\SASHome\SASFoundation\9.4 »). 

 

8.2. Eléments à transmettre au Support Clients 

Si vous rencontrez des problèmes lors de l’utilisation de ces procédures, vous pouvez nous écrire à 

support@sas.com, en attachant à votre message l’erreur reçue et le code utilisé. 
 
 

9. LIENS UTILES 

Pour en savoir plus sur les procédures présentées ci-dessus et les autres procédures disponibles avec 
le module SAS/ETS, nous vous invitons à aller sur la page de la documentation officielle de SAS/ETS 
ou à consulter les nouveautés. 
 

CONCLUSION 

Nous avons parlé dans cet article des principales procédures permettant d’analyser la composante 
temporelle dans vos données mais il existe d’autres horizons plus complexes dans ce module. La 
procédure SIMILARITY, par exemple, permet de travailler sur la similarité entre les séries temporelles 
elles-mêmes (un article entier est consacré à l’application de cette procédure et de ce principe dans 
SAS™ Enterprise Miner). La chose primordiale à retenir est le fait que la notion de temps dans vos 
données est importante et qu’il existe plusieurs moyens différents d’analyser cette composante au 

sein du module SAS/ETS, et très souvent il y a une procédure dédiée pour le faire de la manière dont 
vous le souhaitez, simplement et automatiquement. Les performances de traitement des données 
temporelles sont effectuées de manière automatisée et ont été encore améliorées dans le module 
High-Performance Forecasting. 

